Appendix A1. Power Series and Special Functions

Series Solutions

Maple's **dsolve** procedure can be used to obtain series solutions to differential equations. Add the optional equation

```
type = series
```

The unknown function be inserted first, directly after the IVP (which gnus appear inside a set).

> Example Obtain a series solution to the following differential equation, see Simmons/Krantz, Chapter 4, Section 3, Exercise 1 (a).
>
> $$y'' + x\,y' + y = 0$$
>
> Then obtain the first 6 non-zero terms of the series solution satisfying y(0) = 1, y'(0) = 3. Plot the approximate and the exact solutions. (Think of a mass spring system moving in a fluid that is thickening with time.)

```
> DE := diff(y(x),x,x) + x*diff(y(x),x) + y(x) = 0;
```

$$DE := \frac{d^2}{dx^2}\,y(x) + x\left(\frac{d}{dx}\,y(x)\right) + y(x) = 0 \tag{1}$$

```
> gen_soln := dsolve( DE, y(x), type=series);
```

$$gen\_soln := y(x) = \left(y(0) + D(y)(0)\,x - \frac{1}{2}\,y(0)\,x^2 - \frac{1}{3}\,D(y)(0)\,x^3 + \frac{1}{8}\,y(0)\,x^4 + \frac{1}{15}\,D(y)(0)\,x^5 + O(x^6)\right) \tag{2}$$

Lacking initial conditions, Maple finds the series solution near $x = 0$. This is why the general solution formula is given in terms of the initial values y(0) and D(y)(0). The "order 6" solution is obtained by default. The term

$$O(x^6)$$

represents an error term that, as x approaches 0, also goes to 0 at least as fast as x^6 does. To obtain approximate solution values (or curves) convert the output to a polynomial using the **convert/polynom** procedure..

```
> convert(gen_soln,polynom);
```

$$y(x) = y(0) + x\,D(y)(0) - \frac{1}{2}\,y(0)\,x^2 - \frac{1}{3}\,D(y)(0)\,x^3 + \frac{1}{8}\,y(0)\,x^4 + \frac{1}{15}\,D(y)(0)\,x^5 \tag{3}$$

If initial values are given, they are substituted into the series. To obtain the first 6 non-zero terms ask for the order 7 solution by inserting the equation

```
order = 7
```

```
> soln := dsolve( {DE,y(0)=1,D(y)(0)=3}, y(x), type=series, order=7);
```

$$soln := y(x) = \left(1 + 3\,x - \frac{1}{2}\,x^2 - x^3 + \frac{1}{8}\,x^4 + \frac{1}{5}\,x^5 - \frac{1}{48}\,x^6 + \mathrm{O}(x^7)\right) \tag{4}$$

```
> y_approx := convert(rhs(soln),polynom);
```

$$y\_approx := 1 + 3\,x - \frac{1}{2}\,x^2 - x^3 + \frac{1}{8}\,x^4 + \frac{1}{5}\,x^5 - \frac{1}{48}\,x^6 \tag{5}$$

The next entry generates an exact solution formula. It is rather complicated, given in terms of what is called the ***error function***.: $erf(x) = \frac{2}{\sqrt{\pi}}\int_0^x e^{-t^2}\,\mathrm{d}t$.

```
> exact_soln := dsolve( {DE,y(0)=1,D(y)(0)=3} );
```

$$exact\_soln := y(x) = -\frac{\frac{3}{2}\,\mathrm{I}\,\mathrm{erf}\left(\frac{1}{2}\,\mathrm{I}\,\sqrt{2}\,x\right)\sqrt{\pi}\,\sqrt{2}}{\mathrm{e}^{\frac{1}{2}\,x^2}} + \frac{1}{\mathrm{e}^{\frac{1}{2}\,x^2}} \tag{6}$$

Compare the exact and approximate solution curves plotted below. The approximation is the blue, dashed curve (**linestyle=2**).

```
> plots[setoptions]( tickmarks = [5,5] );
> plot( [rhs(exact_soln), y_approx], x=0..5, 0..5, color=[red,blue],
        linestyle=[1,3]);
```



Want a recursion relation: Use powsolve

When the differential equation has polynomial coefficients and $x = 0$ is not a singular point, the recursion relation can be obtained by applying the **powsolve** procedure in the **powseries** package.

```
> with(powseries);
```

$$[compose, evalpow, inverse, multconst, multiply, negative, powadd, powcos, powcreate, powdiff, powexp, powint, powlog, powpoly, powsin, powsolve, powsqrt, quotient, reversion, subtract, tpsform] \tag{7}$$

```
> pow_soln := powsolve( {DE, y(0)=1, D(y)(0)=3} );
```

$$pow\_soln := \mathbf{proc}(powparm)\ \ldots\ \mathbf{end\ proc} \tag{8}$$

The output is a procedure that can be used to generate power series solutions.

Apply the **tpsform** procedure ("**t**o **p**ower **s**eries **form**") to *pow_soln* as follows to generate the 10th order solution formula

tpsform(pow_soln,x,10)

```
> tpsform(pow_soln,x,10);
```

$$1+3x-\frac{1}{2}x^2-x^3+\frac{1}{8}x^4+\frac{1}{5}x^5-\frac{1}{48}x^6-\frac{1}{35}x^7+\frac{1}{384}x^8+\frac{1}{315}x^9+O\left(x^{10}\right) \qquad (9)$$

Apply *pow_soln* directly to the expression **_k** as shown below to obtain the recursion relation.

```
> pow_soln(_k);
```

$$-\frac{a(\_k-2)}{\_k} \qquad (10)$$

Interpret this output as

$$a_k=-\frac{a_{k-2}}{k}$$

For numbers and graphs, you may be better off with dsolve/numeric

The **dsolve/numeric** and **odeplot** procedures can also be used to obtain a good picture of the exact solution curve.

```
> numeric_soln := dsolve( {DE, y(0)=1, D(y)(0)=3}, y(x), numeric);
```

$$numeric\_soln := \mathbf{proc}\left(x\_rkf45\right)\ \ldots\ \mathbf{end\ proc} \qquad (11)$$

```
> with(plots):
  display( odeplot( numeric_soln, [x,y(x)], x=-0..5, view=0..5,
                    color=red),
           plot( y_approx, x=0..5, 0..5, color=blue, linestyle=3) );
```



The red (solid) curve is the **dsolve/numeric** solution.

Here is now the value given by the numeric solution compares to the exact value, at $x = 3$.

```
> Numeric,numeric_soln(3);
  Exact,eval(exact_soln,x=3): evalf(%);
```

$$Numeric, \left[x=3.,\ y(x)=1.19060960414369086,\ \frac{d}{dx}y(x)=-.571828960889531945\right]$$

$$Exact,\ y(3)=1.190609633 \qquad (12)$$

Special functions

Maple has, as built in procedures, all of the special functions that are found in elementary differential equations textbooks. For example, the *Bessel functions* of the first and second kind of order p are denoted **`BesselJ(p,x)`** and **`BesselY(p,x)`** respectively. See Simmons/Krantz, Chapter 4, Section 4. Their graphs are plotted below.

```
> plot( [BesselJ(0,x),BesselJ(1,x)], x=0..20, linestyle=[1,3]);
```



Bessel functions of the first kind, order 1 and 2 (dashed curve).

```
> plot( [BesselY(0,x),BesselY(1,x)], x=0..20, -1..0.6,
        linestyle=[1,3]);
```



Bessel functions of the second kind, order 1 and 2 (dashed curve).

To see a hotlist of all of the functions that are initially known to Maple (without loading any special packages) go here.

```
> ?inifcn
```

For more information about the functions that are "built in" to Maple go see the **FunctionAdvisor**.

```
> ?FunctionAdvisor
```